# Daisyworld

Study this example to learn about

• Simple agent properties with complex model interactions
• Collecting data with the low-level data collection API
• Diffusion of a quantity in a GridSpace
• the fill_space! function
• represent a space "surface property" as an agent
• counting time in the model and having time-dependent dynamics
• data collection in a mixed-agent model
• performing interactive scientific research

## Overview of Daisyworld

This model explores the Gaia hypothesis, which considers the Earth as a single, self-regulating system including both living and non-living parts.

Daisyworld is filled with black and white daisies. Their albedo's differ, with black daisies absorbing light and heat, warming the area around them; white daisies doing the opposite. Daisies can only reproduce within a certain temperature range, meaning too much (or too little) heat coming from the sun and/or surrounds will ultimately halt daisy propagation.

When the climate is too cold it is necessary for the black daisies to propagate in order to raise the temperature, and vice versa – when the climate is too warm, it is necessary for more white daisies to be produced in order to cool the temperature. The interplay of the living and non living aspects of this world manages to find an equilibrium over a wide range of parameter settings, although with enough external forcing, the daisies will not be able to regulate the temperature of the planet and eventually go extinct.

## Defining the agent types

Daisy has three values (other than the required id and pos for an agent that lives on a GridSpace. Each daisy has an age, confined later by a maximum age set by the user, a breed (either :black or :white) and an associated albedo value, again set by the user. Land represents the surface. We could make Land also have an albedo field, but in this world, the entire surface has the same albedo and thus we make it a model parameter.

Notice that the Land does not necessarily have to be an agent, and one could represent surface temperature via a matrix (parameter of the model). This is done in an older version, see file examples/daisyworld_matrix.jl. The old version has a slight performance advantage. However, the advantage of making the surface composed of agents is that visualization is simple and one can use the interactive application to also visualize surface temperature. It is also available from the Models module as Models.daisyworld.

using Agents
using Statistics: mean

mutable struct Daisy <: AbstractAgent
id::Int
pos::Dims{2}
breed::Symbol
age::Int
albedo::Float64 # 0-1 fraction
end

mutable struct Land <: AbstractAgent
id::Int
pos::Dims{2}
temperature::Float64
end

const DaisyWorld = ABM{<:GridSpace,Union{Daisy,Land}};

## World heating

The surface temperature of the world is heated by its sun, but daisies growing upon it absorb or reflect the starlight – altering the local temperature.

function update_surface_temperature!(pos::Dims{2}, model::DaisyWorld)
ids = ids_in_position(pos, model)
# All grid positions have at least one agent (the land)
absorbed_luminosity = if length(ids) == 1
# Set luminosity via surface albedo
(1 - model.surface_albedo) * model.solar_luminosity
else
# more than 1 agents: daisy exists
# Set luminosity via daisy albedo
(1 - model[ids[2]].albedo) * model.solar_luminosity
end
# We expect local heating to be 80 ᵒC for an absorbed luminosity of 1,
# approximately 30 for 0.5 and approximately -273 for 0.01.
local_heating = absorbed_luminosity > 0 ? 72 * log(absorbed_luminosity) + 80 : 80
# Surface temperature is the average of the current temperature and local heating.
T0 = model[ids[1]].temperature
model[ids[1]].temperature = (T0 + local_heating) / 2
end

In addition, temperature diffuses over time

function diffuse_temperature!(pos::Dims{2}, model::DaisyWorld)
ratio = get(model.properties, :ratio, 0.5) # diffusion ratio
ids = nearby_ids(pos, model)
meantemp = sum(model[i].temperature for i in ids if model[i] isa Land) / 8
land = model[ids_in_position(pos, model)[1]] # land at current position
# Each neighbor land patch is giving up 1/8 of the diffused
# amount to each of *its* neighbors
land.temperature = (1 - ratio) * land.temperature + ratio * meantemp
end

## Daisy dynamics

The final piece of the puzzle is the life-cycle of each daisy. This method defines an optimal temperature for growth. If the temperature gets too hot or too cold, daisies will not wish to propagate. So long as the temperature is favorable, daisies compete for land and attempt to spawn a new plant of their breed in locations close to them.

function propagate!(pos::Dims{2}, model::DaisyWorld)
ids = ids_in_position(pos, model)
if length(ids) > 1
daisy = model[ids[2]]
temperature = model[ids[1]].temperature
# Set optimum growth rate to 22.5 ᵒC, with bounds of [5, 40]
seed_threshold = (0.1457 * temperature - 0.0032 * temperature^2) - 0.6443
if rand(model.rng) < seed_threshold
# Collect all adjacent position that have no daisies
empty_neighbors = Tuple{Int,Int}[]
neighbors = nearby_positions(pos, model)
for n in neighbors
if length(ids_in_position(n, model)) == 1
push!(empty_neighbors, n)
end
end
if !isempty(empty_neighbors)
# Seed a new daisy in one of those position
seeding_place = rand(model.rng, empty_neighbors)
a = Daisy(nextid(model), seeding_place, daisy.breed, 0, daisy.albedo)
end
end
end
end

And if the daisies cross an age threshold, they die out. Death is controlled by the agent_step function

function agent_step!(agent::Daisy, model::DaisyWorld)
agent.age += 1
agent.age >= model.max_age && kill_agent!(agent, model)
end

We also need to define a version for the Land instances (the dynamics of the Land are resolved at model level)

agent_step!(agent::Land, model::DaisyWorld) = nothing

The model step function and agent step functions for Agents.jl to advance Daisyworld's dynamics. Since we have constructed a number of helper functions, these methods are quite straightforward.

function model_step!(model)
for p in positions(model)
update_surface_temperature!(p, model)
diffuse_temperature!(p, model)
propagate!(p, model)
end
model.tick += 1
solar_activity!(model)
end

Notice that solar_activity! changes the incoming solar radiation over time, if the given "scenario" (a model parameter) is :ramp. The parameter tick of the model keeps track of time.

function solar_activity!(model::DaisyWorld)
if model.scenario == :ramp
if model.tick > 200 && model.tick <= 400
model.solar_luminosity += model.solar_change
end
if model.tick > 500 && model.tick <= 750
model.solar_luminosity -= model.solar_change / 2
end
elseif model.scenario == :change
model.solar_luminosity += model.solar_change
end
end

## Initialising Daisyworld

Here, we construct a function to initialize a Daisyworld. We use fill_space! to fill the space with Land instances. Then, we need to know how many daisies of each type to seed the planet with and what their albedo's are. We also want a value for surface albedo, as well as solar intensity (and we also choose between constant or time-dependent intensity with scenario).

import StatsBase
import DrWatson: @dict

function daisyworld(;
griddims = (30, 30),
max_age = 25,
init_white = 0.2, # % cover of the world surface of white breed
init_black = 0.2, # % cover of the world surface of black breed
albedo_white = 0.75,
albedo_black = 0.25,
surface_albedo = 0.4,
solar_change = 0.005,
solar_luminosity = 1.0, # initial luminosity
scenario = :default,
)

space = GridSpace(griddims)
properties = @dict max_age surface_albedo solar_luminosity solar_change scenario
properties[:tick] = 0
# create a scheduler that only schedules Daisies
daisysched(model) = [a.id for a in allagents(model) if a isa Daisy]
model = ABM(
Union{Daisy,Land},
space;
scheduler = daisysched,
properties = properties,
warn = false,
)

# fill model with Land: every grid position has 1 land instance
fill_space!(Land, model, 0.0) # zero starting temperature

# Populate with daisies: each position has only one daisy (black or white)
grid = collect(positions(model))
num_positions = prod(griddims)
white_positions =
StatsBase.sample(grid, Int(init_white * num_positions); replace = false)
for wp in white_positions
wd = Daisy(nextid(model), wp, :white, rand(model.rng, 0:max_age), albedo_white)
end
allowed = setdiff(grid, white_positions)
black_positions =
StatsBase.sample(allowed, Int(init_black * num_positions); replace = false)
for bp in black_positions
wd = Daisy(nextid(model), bp, :black, rand(model.rng, 0:max_age), albedo_black)
end

return model
end

## Visualizing & animating

Lets run the model with constant solar isolation and visualize the result

using InteractiveDynamics, CairoMakie, AbstractPlotting

model = daisyworld()
AgentBasedModel with 1260 agents of type Union{Main.ex-daisyworld.Daisy, Main.ex-daisyworld.Land}
space: GridSpace with size (30, 30), metric=chebyshev and periodic=true
scheduler: daisysched
properties: Dict{Symbol,Any}(:solar_luminosity => 1.0,:max_age => 25,:surface_albedo => 0.4,:solar_change => 0.005,:tick => 0,:scenario => :default)

To visualize we need to define the necessary functions for abm_plot. The daisies will obviously be black or white, but the land will have a color that reflects its temperature, with -50 darkest and 100 ᵒC brightest color

daisycolor(a::Daisy) = a.breed
daisycolor(a::Land) = landcolor[(a.temperature+50)/150]

And we plot daisies as circles, and land patches as squares

daisyshape(a::Daisy) = '❀'
daisysize(a::Daisy) = 13
daisyshape(a::Land) = '■'
daisysize(a::Land) = 15

Notice that we want to ensure that the Land patches are always plotted first.

plotsched = by_type((Land, Daisy), false)

plotkwargs = (ac = daisycolor, am = daisyshape, as = daisysize, scheduler = plotsched)

p, _ = abm_plot(model; plotkwargs...)
p